Double pendulum with springs

Nasser M. Abbasi

January 22, 2020 Compiled on January 31, 2024 at 4:28am



Figure 1: Geometry of the problem

Assuming both springs have the same relaxed length of L. Starting by finding the Lagrangian $\mathcal{L}=T-V$. For m_1

$$T_1=\frac{1}{2}m_1\left(\dot{x}_1^2+\left((L+x_1)\,\dot{\theta}_1\right)^2\right)$$
$$V_1=-m_1g(L+x_1)\cos\theta_1+\frac{1}{2}k_1x_1^2$$

And for m_2

$$\begin{aligned}T_2=&\frac{1}{2}m_2\left((\dot{x}_2+\dot{x}_1\cos(\theta_1-\theta_2))^2+(\dot{x}_1\sin(\theta_1-\theta_2))^2\right)\\&+\frac{1}{2}m_2\left(\left((L+x_2)\,\dot{\theta}_2+(L+x_1)\,\dot{\theta}_1\cos(\theta_1-\theta_2)\right)^2+\left((L+x_1)\,\dot{\theta}_1\sin(\theta_1-\theta_2)\right)^2\right)\\V_2=&-m_2g((L+x_1)\cos\theta_1+(L+x_2)\cos\theta_2)+\frac{1}{2}k_2x_2^2\end{aligned}$$

Hence

$$\mathcal{L}=(T_1+T_2)-(V_1+V_2)$$

There are 4 generalized coordinates, $x_1, x_2, \theta_1, \theta_2$. Now Mathematica is used to obtain the four equations of motion to help with the algebra. Once $x_1, x_2, \theta_1, \theta_2$ are solved for, the position of each mass m_1, m_2 is fully known at each time instance, and each mass motion can be animated. The four equations of motion are

$$
\begin{aligned}
\frac{d}{dt}\left(\frac{\partial \mathcal{L}}{\partial \dot{x}_1}\right) - \frac{\partial \mathcal{L}}{\partial x_1} &= 0 \\
\frac{d}{dt}\left(\frac{\partial \mathcal{L}}{\partial \dot{x}_2}\right) - \frac{\partial \mathcal{L}}{\partial x_2} &= 0 \\
\frac{d}{dt}\left(\frac{\partial \mathcal{L}}{\partial \dot{\theta}_1}\right) - \frac{\partial \mathcal{L}}{\partial \theta_1} &= 0 \\
\frac{d}{dt}\left(\frac{\partial \mathcal{L}}{\partial \dot{\theta}_2}\right) - \frac{\partial \mathcal{L}}{\partial \theta_2} &= 0
\end{aligned}
$$

The rest is done using Mathematica to help with the algebra

```
ClearAll[x1, x2, θ1, θ2, t]
T1 = 1/2 m1 (x1'[t]^2 + ((L + x1[t]) θ1'[t])^2);
V1 = -m1 g (L + x1[t]) Cos[θ1[t]] + 1/2 k1 x1[t]^2;
T2 =
    1/2 m2 ((x2'[t] + x1'[t] Cos[θ1[t] - θ2[t]])^2 + (x1'[t] Sin[θ1[t] - θ2[t]])^2) +
      1/2 m2 (((L + x2[t]) θ2'[t] + ((L + x1[t]) θ1'[t] Cos[θ1[t] - θ2[t]]))^2 +
            ((L + x1[t]) θ1'[t] Sin[θ1[t] - θ2[t]])^2);
V2 = -m2 g ((L + x1[t]) Cos[θ1[t]] + (L + x2[t]) Cos[θ2[t]]) + 1/2 k2 x2[t]^2;


(*Lagrangian*)

In[71]:= (lag = (T1 + T2) - (V1 + V2)) // Simplify
```

Out[71]=

$$\frac{1}{2}\Big(-k1\,x1[t]^2 + 2\,g\,m1\,\mathrm{Cos}[\theta 1[t]]\,(L + x1[t]) - k2\,x2[t]^2 + 2\,g\,m2\,(\mathrm{Cos}[\theta 1[t]]\,(L + x1[t]) + \mathrm{Cos}[\theta 2[t]]\,(L + x2[t])) + m2\,(x1'[t]^2 + 2\,\mathrm{Cos}[\theta 1[t] - \theta 2[t]]\,x1'[t]\,x2'[t] + x2'[t]^2) + m1\,(x1'[t]^2 + (L + x1[t])^2\,\theta 1'[t]^2) + m2\,\big(\mathrm{Sin}[\theta 1[t] - \theta 2[t]]^2\,(L + x1[t])^2\,\theta 1'[t]^2 + (\mathrm{Cos}[\theta 1[t] - \theta 2[t]]\,(L + x1[t])\,\theta 1'[t] + (L + x2[t])\,\theta 2'[t])^2\big)\Big)$$

```
(*x1 *)

In[72]:= (eq1 = D[D[lag, x1'[t]], t] - D[lag, x1[t]] == 0) // Simplify

(*x2*)

In[73]:= (eq2 = D[D[lag, x2'[t]], t] - D[lag, x2[t]] == 0) // Simplify

(*theta 1*)

In[75]:= (eq3 = D[D[lag, θ1'[t]], t] - D[lag, θ1[t]] == 0) // Simplify

In[76]:= (eq4 = D[D[lag, θ2'[t]], t] - D[lag, θ2[t]] == 0) // Simplify

(*Numerically solve the equations of motion*)

pars = {L -> 1, m1 → 1, m2 → 2, g → 9.8, k1 → 10, k2 → 30};
ic = {θ1[0] == 5 Degree , θ1'[0] == 0, θ2[0] == 3 Degree,
     θ2'[0] == 0, x1[0] == 0, x1'[0] == 0, x2[0] == 0, x2'[0] == 0};
eqs = Flatten[{eq1, eq2, eq3, eq4}] /. pars

In[122]:= numericalSolution = First@NDSolve[{eqs, ic}, {x1, x2, θ1, θ2}, {t, 0, 20}];
```

In[120]:=
```
Plot[Evaluate[({θ1[t], θ2[t]} /. numericalSolution) * 180 / Pi],
  {t, 0, 20}, PlotRange → All, AxesLabel → {"time", "Angle (in degree)"},
  ImageSize → 400, PlotLegends → {"mass 1", "mass 2"}]
```

Out[120]=



In[121]:=
```
Plot[Evaluate[{x1[t], x2[t]} /. numericalSolution], {t, 0, 20},
  PlotRange → All, AxesLabel → {"time", "spring extensions in meters"},
  ImageSize → 400, PlotLegends → {"mass 1", "mass 2"}]
```

Out[121]=